<commit_before>{ stdenv, fetchurl, ffmpeg, autoconf, automake, libtool, pkgconfig, log4cpp
, pango, cairo, python, libjpeg, ffms
, enableQt ? true, qt4}:

stdenv.mkDerivation rec {
  name = "avxsynth-4.0-e153e672bf";

  src = fetchurl {
    url = https://github.com/avxsynth/avxsynth/tarball/e153e672bf;
    name = "${name}.tar.gz";
    sha256 = "16l2ld8k1nfsms6jd9d9r4l247xxbncsak66w87icr20yzyhs14s";
  };

  buildInputs = [ ffmpeg autoconf automake libtool pkgconfig log4cpp pango cairo python
    libjpeg ffms ]
    ++ stdenv.lib.optional enableQt qt4;

  preConfigure = "autoreconf -vfi";

  meta = {
    homepage = https://github.com/avxsynth/avxsynth/wiki;
    license = stdenv.lib.licenses.gpl2Plus;
    maintainers = with stdenv.lib.maintainers; [viric];
    platforms = with stdenv.lib.platforms; linux;
  };
}<commit_msg>avxsynth: refactor & 2013-05-10 -> 2015-04-07<commit_after>{ stdenv, fetchFromGitHub, autoreconfHook, pkgconfig
, cairo, ffmpeg, ffms, libjpeg, log4cpp, pango
, avxeditSupport ? false, qt4 ? null
}:

let
  inherit (stdenv.lib) enableFeature optional;
in

stdenv.mkDerivation rec {
  name = "avxsynth-${version}";
  version = "2015-04-07";

  src = fetchFromGitHub {
    owner = "avxsynth";
    repo = "avxsynth";
    rev = "80dcb7ec8d314bc158130c92803308aa8e5e9242";
    sha256 = "0kckggvgv68b0qjdi7ms8vi97b46dl63n60qr96d2w67lf2nk87z";
  };

  configureFlags = [
    "--enable-autocrop"
    "--enable-framecapture"
    "--enable-subtitle"
    "--enable-ffms2"
    (enableFeature avxeditSupport "avxedit")
    "--with-jpeg=${libjpeg}/lib"
  ];

  nativeBuildInputs = [ autoreconfHook pkgconfig ];

  buildInputs = [ cairo ffmpeg ffms libjpeg log4cpp pango ]
    ++ optional avxeditSupport qt4;

  meta = with stdenv.lib; {
    description = "A script system that allows advanced non-linear editing";
    homepage = https://github.com/avxsynth/avxsynth;
    license = licenses.gpl2Plus;
    maintainers = with maintainers; [ codyopel viric ];
    platforms = platforms.linux;
  };
}
{'fraction_non_alphanumeric': 0.12282446544007956, 'fraction_numerical': 0.06166086524117355, 'mean_word_length': 3.813397129186603, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 3, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6977144', 'n_tokens_mistral': 864, 'n_tokens_neox': 774, 'n_words': 145}